Using did_multiplegt_dyn in Stata to Estimate Event-Study Effects in Complex Designs: Four Examples Based on Real Datasets

Bingxue Li & Clément de Chaisemartin & ERC Really Credible Team
chaisemartin.packages@gmail.com

July 3, 2025

1 Introduction

This document provides examples where the Stata command did_multiplegt_dyn, which computes the estimators proposed in De Chaisemartin and d'Haultfoeuille (2024), is used on real datasets to estimate event-study effects in complex designs. The first example has a binary treatment that can turn on an off. The second example has a continuous absorbing treatment. The third example has a discrete multivalued treatment that can increase or decrease multiple times over time. The fourth example has two, binary and absorbing treatments, where the second treatment always happens after the first. To access the datasets used in the following examples, first run:[1]

```
ssc install did_multiplegt_dyn
net get did_multiplegt_dyn
```

2 Deryugina (2017a): On and Off Binary Treatment

2.1 Research Question

Example based on *The Fiscal Cost of Hurricanes: Disaster Aid versus Social Insurance* by Deryugina (2017a):

> How do hurricanes affect non-disaster government transfer programs in the United States?

2.2 Data

```
use Deryugina_2017.dta, clear
```

The data come from the publicly available dataset in the replication package of Deryugina (2017b). The data is a panel of US counties at the yearly level. The variable hurricane is the binary treatment. The outcome is log_curr_trans_ind_gov_pc, the log of non-disaster government transfer received by county g in year t, divided by that county's population.

2.3 On-and-off one-shot design: counties are treated for at most one period

```
egen total_hurricane = total(hurricane), by(county_fips)
sum total_hurricane
```

We sum the treatment values across years by county, and the total treatment takes the value of 0 or 1 for all counties. Therefore, we are working with a non-absorbing binary treatment, that can turn on and off. Moreover, counties are treated for at most one year, so we are in a so-called "one-shot" design (De Chaisemartin and d'Haultfoeuille, 2024).

[1] You can also find the data and the do-file on our GitHub page.

2.4 Estimate Treatement Effects with `did_multiplegt_dyn`

The author uses an event-study two-way fixed effects regression to estimate hurricanes' effects on non-disaster government transfer per capita for 11 years after a hurricane. Instead, we use `did_multiplegt_dyn` to estimate the same effects with an estimator robust to heterogeneous effects across counties and over time. We also compute eleven pre-trends estimators.

```
* Call did_multiplegt_dyn
did_multiplegt_dyn log_curr_trans_ind_gov_pc county_fips year hurricane,
    effects(11) placebo(11) cluster(county_fips)
```



Figure 1: `did_multiplegt_dyn`: Effects of Hurricane Hits on Government Transfers Per Capita

The event-study estimators suggest that hurricanes increase non-disaster government transfer per capita by 2% in the short run, and by 4% in the long run. In a one-shot treatment design, under the parallel-trends assumption event-study effects significantly different from zero at $\ell \geq 2$ imply that we can reject the no-dynamic or no-carryover effect hypothesis (Liu et al., 2024). Intuitively, as treated groups revert back to being untreated immediately after receiving the treatment, any remaining treatment effect in subsequent periods must come from effects of lagged treatments on the outcome.

The previous discussion relies on no-anticipation and parallel-trends assumptions. However, the pre-trend estimators are statistically significant, and the event-study estimators could just be the continuation of differential trends between treated and untreated counties that seem to start 4 (-3-1) periods before the treated counties get treated.

To address this potential violation of the no-anticipation and parallel-trends assumptions, we follow the author and control for several county-level characteristics. These are all time-invariant variables, like the counties' log population in 1969, that capture demographic and economic conditions at the start of the panel. Following Section 4.1 of de Chaisemartin and D'Haultfœuille (2023), the variables inputted to the `controls()` option of `did_multiplegt_dyn` have to be defined as $X_g \times t$, to allow for linear trends interacted with the covariates.

```
* Time-invariate covariates: Adding covariates*year
local vars coastal land_area1970 log_pop1969 frac_young1969 frac_old1969
    frac_black1969 log_wage_pc1969 emp_rate1969
foreach v of local vars {
    gen `v'_year = `v' * year
    local controls `v'_year
}
* Call did_multiplegt_dyn with controls
did_multiplegt_dyn log_curr_trans_ind_gov_pc county_fips year hurricane,
    effects(11) placebo(11) controls(`controls') cluster(county_fips)
```



Figure 2: `did_multiplegt_dyn`: Effects of Hurricane Hits on Government Transfers Per Capita, with county-level controls

After including those controls, the pre-trends estimators are slightly less significant, but the differential trend starting 4 periods before treatment persists.

3 East et al. (2023): Treatment Continuously Distributed in Period 1

3.1 Research Question

Example based on *Multigenerational Impacts of Childhood Access to the Safety Net: Early Life Exposure to Medicaid and the Next Generation's Health* by East et al. (2023):

How does prenatal Medicaid expansion affect birth weight for the directly treated generation?

3.2 Data

```
use east_et_al_2023.dta, replace
```

The data come from the publicly available dataset in the replication package of East et al. (2022). The data is a panel of US states at the yearly level. The variable `newsimeli` is the treatment. To construct it, we start from the simulated prenatal Medicaid eligibility rate in each state and year, constructed by East et al. (2023). This simulated eligibility rate is designed to isolate changes driven solely by Medicaid policy rules, independent of changes in states' demographic and economic characteristics. The authors classify 28 states, in which eligibility was stagnant between 1975 and 1979, and that later experienced a large positive shock, as their treatment group. The remaining 22 states are the control group. We construct a continuous treatment variable in the spirit of their classification. For the 22 control states, `newsimeli` is equal, throughout the study period, to the average simulated eligibility rate from 1975 to 1979. For the 28 treated states, `newsimeli`'s definition is the same before the year of the large positive shock identified by the authors. After that year, `newsimeli` is equal to the simulated eligibility rate in the year of the shock. Thus, we construct a treatment which varies across states at every time period, to reflect variation in Medicaid coverage across states. At the same time, for each state the treatment changes at most once over time: identification will only leverage the large shocks in eligibility rules identified by the authors. The outcome is `lbw_detrend81`, the percentage of infants born in state g and year t who were below the low birthweight threshold, netted from a state-specific pre-treatment linear trend.

3.3 Design: the treatment takes several different values in period 1

```
tab newsimeli if dob_y_p==1975
```

3.4 Estimate Treatement Effects with did_multiplegt_dyn

We first use did_multiplegt_dyn to estimate non-normalized event-study effects of experiencing a large increase of prenatal Medicaid eligibility rate on states' low-birth-weight (LBW) rates. We compute five event-study estimators and five pre-trends estimators. Because the period-one treatment is continuous, we specify the continuous(1) option, thus assuming that states's counterfactual trends without a treatment change is linear in their baseline treatment. See Section 1.10 of the Web Appendix of De Chaisemartin and d'Haultfoeuille (2024) for further details. With the continuous option, the analytic ses computed by the command may not always be reliable, so we use the boostrap instead. Finally, we control for the same time-varying covariates as the authors, and we weight the estimation by the number of births in state g and year t. We also use the effects_equal("all") option to perform an F-test of the null hypothesis that all five estimated dynamic effects are equal.

```
global stcontrols stmarried stblack stother sthsdrop ///
sthsgrad stsomecoll pop0_4 pop5_17 pop18_24 pop25_44 ///
pop45_64 urate incpc maxafdc abortconsent abortmedr
did_multiplegt_dyn lbw_detrend81 plborn dob_y_p newsimeli, effects(5) placebo
    (5) continuous(1) bootstrap(50,1) controls($stcontrols) weight(births)
    effects_equal("all")
```



Figure 3: Effects of a Large Increase in Prenatal Medicaid Eligibility on Low-Birth-Weight Rate

Results suggest that experiencing a large increase of prenatal Medicaid eligibility rate significantly decreases states' LBW rates. Pre-trend estimators are small and insignificant, thus suggesting that the no-anticipation and parallel-trends assumptions might be plausible in this application.

Instead of estimating a reduced-form effect of "experiencing a large increase of prenatal Medicaid eligibility rate", one might be interested in estimating how the LBW rate responds to a 1 percentage point increase in the prenatal Medicaid eligibility rate. For that purpose, we now use did_multiplegt_dyn to estimate normalized event-study effects.

```
did_multiplegt_dyn lbw_detrend81 plborn dob_y_p newsimeli, effects(5) placebo
    (5) continuous(1) bootstrap(50,1) controls($stcontrols) weight(births)
    normalized normalized_weights effects_equal("all")
```



Figure 4: Effects of a One Percentage Point Increase in Prenatal Medicaid Eligibility on Low-Birth-Weight Rate

The first normalized event-study estimator indicates that increasing the eligibility rate by one percentage point reduces the LBW rate by one percentage point. The second normalized event-study estimator is a weighted average, with weights 1/2 and 1/2, of the effects of the current and lagged eligibility rates on the LBW rate of state g and year t. Again, a weighted average of the effects of increasing the current and lagged eligibility rates by one percentage point leads to a reduction of the LBW rate by one percentage point. The third normalized event-study estimator is a weighted average, with weights 1/3, 1/3, and 1/3, of the effects of the current and first and second lags of the eligibility rate, etc. Pre-trend estimators are normalized by the same quantity as the event-study estimators, to ensure that they can give researchers a sense of the magnitude of the potential bias in the normalized event-study estimators coming from differential trends.

4 Gentzkow et al. (2011a): On-and-off Discrete Treatment

4.1 Research Question

Example based on *The effect of newspaper entry and exit on electoral politics* by Gentzkow et al. (2011a):

What is the effect of daily newspapers on voter turnout?

4.2 Data

```
use gentzkowetal_didtextbook.dta, clear
```

The data come from the publicly available dataset in the replication package of Gentzkow et al. (2011b). The data is an imbalanced panel of 1,195 US counties at the presidential-election-year level, from the 1872 to the 1928 election. The variable `numdailies`, the number of English-language daily newspapers in circulation in county g and election-year t. The outcome is `prestout`, the turnout rate in county g and election-year t.

4.3 Design: On-and-off Discrete Treatment

We first examine the values that the treatment variable `numdailies` takes in the first period of observation.

```
xtset cnty90 year
sort cnty90 year
bysort cnty90 (year): gen first_numdailies = numdailies[1]
tab first_numdailies
```

It takes the following values: $\{0, 1, 2, 3, 4, 5, 6, 7, 9, 11, 12, 16, 33\}$. The treatment is a multivalued discrete variable, even in the first election-year in the data.

We then compute the first-difference of numdailies.

```
gen d_numdailies = .
bysort cnty90 (year): replace d_numdailies = numdailies - numdailies[_n-1]
tab d_numdailies
```

d_numdailies takes positive and negative values: counties can experience both increases and decreases in their number of newspapers.

Finally, we compute the number of times a county experiences a change in numdailies over the duration of the panel.

```
gen switch = (d_numdailies!=0&d_numdailies!=.)
bys cnty90: egen switches=total(switch)
tab switches
```

The vast majority of counties experience more than one change in numdailies. 19% of counties experience 7 changes or more.

4.4 Estimate Treatement Effects with did_multiplegt_dyn

We start by computing four non-normalized event-study estimators and four pre-trends estimators.

```
did_multiplegt_dyn prestout cnty90 year numdailies, effects(4) placebo(4)
   effects_equal("all")
```



Figure 5: did_multiplegt_dyn: Non-normalized Effects of Newspaper Entries/Exits on Voters Turnout

Being exposed to a weakly larger number of newspapers for one electoral cycle increases turnout, and the effect is statistically significant. That effect can be estimated for 1,119 out of the 1,195 counties in the data: 34 counties never experience a change in their number of newspapers, and 42 counties that do experience a change cannot be matched with a not-yet-switcher with the same number of newspapers at baseline. Being exposed to a weakly larger number of newspapers for two, three, and four electoral cycle also significantly increases turnout. Effects increase with exposure length, but one cannot reject the null that all effects are equal. As ℓ increases, effects mechanically apply to fewer and fewer counties, but the effect after four electoral cycles still applies to 917 counties. Pre-trend estimates are small and individually and jointly insignificant. However, their confidence intervals are quite large. The confidence interval of the fourth pre-trend estimator is already quite large, but that of the fifth one (not shown) is substantially larger, so we have very little power to detect differential trends over more than five election cycles. This is why we only report four placebo and four event-study estimators.

Next, we call the `design([float], string)` option. This option reports switchers' period-one and subsequent treatments, thus helping the analyst understand the treatment paths whose effect is aggregated in the non-normalized event-study effects. When the number of treatment paths is large, one may specify a number x included between 0 and 1 in the float argument, so that the command only reports the most common paths accounting for x% of the treatment effects aggregated in the non-normalized event-study effects.

```
did_multiplegt_dyn prestout cnty90 year numdailies, effects(2) design(0.8,
    console)
```

The interpretation goes as follows: 32% of the effects aggregated by the second event-study estimator are effects of having 1 newspaper and 1 lagged newspaper instead of having 0 newspaper and 0 lagged newspaper, 18% are effects of having 0 newspaper and 1 lagged newspaper instead of having 0 newspaper and 0 lagged newspaper, and 12% are effects of having 2 newspapers and 1 lagged newspaper instead of having 0 newspaper and 0 lagged newspaper.

Next, we compute normalized event-study effects.

```
did_multiplegt_dyn prestout cnty90 year numdailies, effects(4) placebo(4)
    normalized normalized_weights effects_equal("all")
```



Figure 6: `did_multiplegt_dyn`: Normalized Effects of Newspaper Entries/Exits on Voters Turnout

Normalized event-study estimates are decreasing with ℓ, but one cannot reject the null that all effects are equal (p-value=0.17). The table "Weights on treatment lags", produced when the `normalized_weights` option is specified, shows that the first event-study estimate is an effect of contemporaneous newspapers on turnout. The second normalized effect is a weighted average of the effects of contemporaneous newspapers and of the first lag of newspapers on turnout, with approximately equal weights. The third normalized effect is a weighted average of the effects of contemporaneous newspapers, the first lag of newspapers, and the second lag of newspapers, with approximately equal weights, etc. Then, the fact that normalized event-study estimates are decreasing with ℓ may suggest that lagged newspapers have a smaller effect on turnout than contemporaneous newspapers.

Finally, we run a joint test that the first lagged treatment has no effect on the outcome, and that treatment effects are constant over time. For that purpose, we estimate the first two non-normalized event-study effects, with the same `same_switchers` option to restrict the sample to counties for which both effects can be estimated, and further restricting the sample to counties whose number of newspapers does not change in the election cycle just after their newspapers changed for the first time or whose number of newspapers has not changed yet (`year<=first_change| same_treat_after_first_change==1`). Then, the test evaluates whether the two non-normalized effects are equal within the selected subsample.

```
did_multiplegt_dyn prestout cnty90 year numdailies
```

```
if year<=first_change|same_treat_after_first_change==1,
effects(2) effects_equal("all") same_switchers graph_off
```

With a p-value of 0.83, the test is not rejected, thus suggesting that the first lag of newspapers does not affect current turnout.

5 Hollingsworth et al. (2022b): Two Consecutive Binary and Absorbing Treatments

5.1 Research Question

Example based on *Comparative Effects of Recreational and Medical Marijuana Laws on Drug Use among Adults and Adolescents* by Hollingsworth et al. (2022b):

How do medical and recreational marijuana laws differentially affect marijuana use among adults and adolescents?

5.2 Data

```
use hollingsworth_et_al_2022.dta, clear
```

The data come from the publicly available dataset in the replication package of Hollingsworth et al. (2022a). The data is a panel of US states at the yearly level. The treatment variables `mm` and `rm`, are respectively equal to 1 if state g at year t has legalized Marijuana consumption for medical and recreational purposes. The outcome variable is `ln_mj_use_365`, the log of marijuana consumption in state g and year t.

5.3 Design: Two Consecutive Binary and Absorbing Treatments

First, we verify that the two treatments are binary and absorbing.

```
xtset state year
gen d_mm = D.mm
gen d_rm = D.rm
tab mm
tab rm
tab d_mm
tab d_rm
```

The treatments take values 0 or 1. The first-difference of the treatments take values of 0 or 1. Therefore, the two treatments are binary and absorbing.

Next, we verify that the medical legalization treatment always happens before the recreational legalization treatment.

```
gen mm_rm_timing = mm-rm
tab mm_rm_timing
```

The difference between mm and rm is always positive, thus showing that the medical legalization treatment always happens before the recreational legalization treatment.

As we have two consecutive binary and absorbing treatments, we follow Section 8.3.4.6 of de Chaisemartin and D'Haultfœuille (2023) to separately estimate the effect of each treatment.

5.4 Estimate Effects of the First Treatment with `did_multiplegt_dyn`

First, we estimate the effect of medical marijuana laws. For that purpose, we just restrict the sample to all state×year (g, t) such that state g has not passed a recreational law yet in year t.

```
did_multiplegt_dyn ln_mj_use_365 state year mm if rm==0, placebo(3) effects(3)
```



Figure 7: `did_multiplegt_dyn`: Effects of Medical Marijuana Laws on Marijuana Use

Results suggest that medical marijuana laws increase marijuana consumption by 3-4%, though effects are marginally significant. Pre-trends are small and insignificant, but their confidence intervals are quite large.

5.5 Estimate Effects of the Second Treatment with `did_multiplegt_dyn`

Next, we estimate the additional effect of a recreational marijuana law. We can estimate that effect under the assumption that the effect of one additional period of exposure to a medical law is the same in every group. This assumption enables us to disentangle the effect of the second treatment from the state-specific incremental impact of the medical law treatment for an additional period. Under that assumption, estimation goes as follows. First, we restrict the sample to (g, t) cells such that state g has implemented a medical law at t. Then, we record the date when that medical law was passed, which will be inputted in the `trends_nonparam()` option when estimating the effects of recreational laws. This ensures that the estimators will compare outcome evolutions between states that adopt the recreational law and those that do not, conditional on having adopted the medical law in the same year.

```
egen fg1 = min(cond(mm == 1, year, .)), by(state)
did_multiplegt_dyn ln_mj_use_365 state year rm if mm ==1, placebo(3) effects
   (3) trends_nonparam(fg1)
```



Figure 8: `did_multiplegt_dyn`: Effects of Recreational Marijuana Laws on Marijuana Use.

Results suggest that recreational laws increase consumption by 6% in the short run, and by 16% in the long run, though effects are marginally significant. Effects apply to a small number of states (the effect after one year is estimated for seven states, that after two years is estimated for five states, that after three years is estimated for two states).[2] Then, the asymptotic approximations underlying the command's analytic standard errors may not be reliable. Somewhat reassuringly, using the bootstrap yields similar standard errors. Pre-trends are small and insignificant, but their confidence intervals are quite large.

[2]This is not due to controlling for the adoption date of medical marijuana laws, effects are estimated for the same number of states without that control.

References

De Chaisemartin, C. and d'Haultfoeuille, X. (2024). Difference-in-differences estimators of intertemporal treatment effects. *Review of Economics and Statistics*, pages 1–45.

de Chaisemartin, C. and D'Haultfœuille, X. (2023). Credible answers to hard questions: Differences-in-differences for natural experiment. https://papers.ssrn.com/sol3/papers.cfm?abstract_id=4487202.

Deryugina, T. (2017a). The fiscal cost of hurricanes: Disaster aid versus social insurance. *American Economic Journal: Economic Policy*, 9(3):168–98.

Deryugina, T. (2017b). Replication data for: The fiscal cost of hurricanes: Disaster aid versus social insurance. Nashville, TN: American Economic Association [publisher], 2017. Ann Arbor, MI: Inter-university Consortium for Political and Social Research [distributor], https://doi.org/10.3886/E114620V.

East, C., Miller, S., Page, M., and Wherry, L. (2022). Data and code for: Multi-generational impacts of childhood access to the safety net: Early life exposure to medicaid and the next generation's health. Nashville, TN: American Economic Association [publisher], 2022. Ann Arbor, MI: Inter-university Consortium for Political and Social Research [distributor], https://doi.org/10.3886/E170801V1.

East, C. N., Miller, S., Page, M., and Wherry, L. R. (2023). Multigenerational impacts of childhood access to the safety net: Early life exposure to medicaid and the next generation's health. *American Economic Review*, 113(1):98–135.

Gentzkow, M., Shapiro, J. M., and Sinkinson, M. (2011a). The effect of newspaper entry and exit on electoral politics. *American Economic Review*, 101(7):2980–3018.

Gentzkow, M., Shapiro, J. M., and Sinkinson, M. (2011b). Replication data for: The effect of newspaper entry and exit on electoral politics. Nashville, TN: American Economic Association [publisher], 2011. Ann Arbor, MI: Inter-university Consortium for Political and Social Research [distributor], https://doi.org/10.3886/E112472V1.

Hollingsworth, A., Wing, C., and Bradford, A. (2022a). Replication package for: Comparative effects of recreational and medical marijuana laws on drug use among adults and adolescents. Journal of Law and Economics. Archived at Zenodo, https://doi.org/10.5281/zenodo.6588013.

Hollingsworth, A., Wing, C., and Bradford, A. C. (2022b). Comparative effects of recreational and medical marijuana laws on drug use among adults and adolescents. *Journal of Law and Economics*, 65(3):515–554.

Liu, L., Wang, Y., and Xu, Y. (2024). A practical guide to counterfactual estimators for causal inference with time-series cross-sectional data. *American Journal of Political Science*, 68(1):160–176.